
Abstract. A heterojunction phototransistor (HPT) is integrated in series with a light-emitting diode to form an optoelectronic device called Light Amplifying Optical Switch. This device has been shown to switch from low current region to high current region at a voltage called breakover voltage. This switching has been attributed to the presence of optical and electrical feedback mechanisms as well as to gain nonlinearity. A circuit model for such device that took into account different factors affecting the switching behavior has been proposed previously. In this paper, a modification to the previously published circuit is made and a spice model is presented for the modified circuit and hence for the LAOS. The spice results presented here are compared to published results and showed a good qualitative agreement. In addition, several factors that affect the switching behavior of such device are analyzed and a good agreement with published results is achieved. The spice model presented here can be very helpful to engineering students in studying optoelectronic circuits and systems that use LAOS.

Introduction

Optoelectronic devices that provide optical gain are very important for opto-electronic integrated circuits implementation.The light amplifying optical switch (LAOS) proposed by Benking [1,2] and Sasaki [3-6] can be considered as a good candidate for the realization of two-dimensional optical arrays which require cascadable devices with optical gain [7]. This device consists of a heterojunction phototransistor (HPT) integrated with a light emitting diode (LED) as shown in Fig. 1. The HPT acts as an optical detector with gain while the LED provides the output optical signal in response to the HPT optical input. This device has been shown to switch from a low current state to a high current state [3,8]. This switching mechanism resulted from the fact that some of the light emitted from the LED is re-absorbed by the HPT which provide an optical positive feedback. Also, some carriers may be injected from the LED to the base of the transistor, providing an electrical positive feedback. When those positive feedbacks exceed a certain threshold, the LAOS switches to a low impedance state. This device has 

found several applications such as implementation of optical inverters[9], NOR[10], AND[11], and NAND[12] gates.

The switching behavior of the LAOS has been exclusively studied by Feld et al. [8] and Beytte et al. [9] and was attributed to the combination of non-linearity in the HPT and to an electrical and optical feedback between the LED and the HPT. An equivalent circuit for the LAOS is presented there.

The purpose of this paper is to use the widespread PSPICE software to simulate this device in an easy manner. This will be very helpful either for optoelectronic engineers or for educational purposes and will add to the eforts to extend the use of PSPICE to simulate optoelectronic devices either as discrete devices or when integrated with conventional electronic devices. 

The Simulated Circuit

The circuit used here is the same circuit proposed by Feld [13] with a slight difference that is the optical current is set between the collector and the base since it is due to the light absorbed there which also results in better accuracy . The modified circuit is shown in Fig. 2. In the circuit, Iin is the equivalent base current produced by the input light.  Iel and Iop are the equivalent electrical and optical feedback currents, respectively.  The current source as well as the resistor in the output is used to model the Early effect and the leakage effect, respectively. For the development of the circuit, the reader is referred to Ref. [8,13].

To simulate the heterojunction bipolar transistor, the Ebers–Moll model proposed by Lundstrom [14] is utilized. In this model the currents are given by:

 IC = qA[FEN – FBR – FCP]






        (1)

IE = qA[FEN + FEP]






        (2)

where:

q  represents the electron charge

A is the device area

FEN represents electrons injected from emitter to base and given by [14]:

where,

VBE and VBC represent the Base-Emitter and Base-Collector voltages respectively 

SN represents the junction velocity, which is given by:

where 

-directed thermal velocity , vjp is the potential drop across the p-side of the junction and ∆Ec is the conduction band discontinuity.

DB and WB are, respectively, the diffusion constant in the base and the base width and  nOB is the equilibrium minority carrier concentration in the base.

Furthermore,

FBR is the base recombination flux given by [10]:

(7)

where B is the lifetime of minority carriers in the base.

The hole fluxes presented in eqs.1 and 2 are given by [14]:

        (8)

and

where

and

                    (11)

DE ( C ) and WE (C) are the diffusion constant in the emitter (collector) and the emitter (collector) width respectively,  POE (C) is the equilibrium minority carrier concentration in the emitter.

Although a more accurate LED model can be easily incorporated, the spice model for the diode is used here for the sake of simplicity.

Analysis Results

The input file of the simulated circuit is shown in Fig. 3, which also shows the parameters used for the HPT.  Usually the parameters are stored in a library file; however, they are listed here along with the circuit file so they can be easily followed.  Furthermore, in order to redefine the transistor terminal currents, the Analog Behavior Modeling option given by PSPICE is used.

*test file for LAOS

* The Parameters 

.PARAM PB=5E19


* base doping*

.PARAM NE=1E18


*emitter doping*

.PARAM NC=2E16


*collector doping*

.PARAM WE=5E-6


* emitter width*

.PARAM WB=1E-5


*base width*

.PARAM DB=50



*base diffusion constant*

.PARAM DE=5



* emitter diffusion constant*

.PARAM DC=30



* collector diffusion constant*

.PARAM WC=8E-5


* collector width*

.PARAM Q=1.6E-19

.PARAM A=1E-4

.PARAM TAUB=1E-9


* base life time*

.PARAM NIB=4E6


* intrinsic carrier concentration in the base*

.PARAM NIE=1.9E3


* intrinsic carrier concentration in the emitter*

.PARAM SEN1=7E2

.PARAM SCN1=1E7

.PARAM SEP1=4E6

.FUNC LB() {(DB*TAUB)^(.5)}

* diffusion length in the base*

.PARAM SCP1=9.4E4

.FUNC POE() {(NIE*NIE/NE)}

.FUNC NOB() {(NIB*NIB/PB)}

.FUNC POC() {NIB*NIB/NC}

.FUNC SEN() {SEN1*WB/DB}

* equation 4

.FUNC SCN() {SCN1*WB/DB}

* equation 5

.FUNC SEP() {SEP1*WE/DE}

* equation 10

.FUNC SCP() {SCP1*WC/DC}

* equation 11

.FUNC F1()  {(DB*NOB)/(WB*(1+1/SEN()+1/SCN()))}
* first part of equation 3

.FUNC F2()  {(DE*POE)/(WE*(1+1/SEP()))}

* first part of equation 8

.FUNC IES() {Q*A*(F1()+F2())}



* equation  2

.FUNC F3()  {(2*(SEN()+SCN()*SEN()))/(SEN()+SCN()+(SCN()*SEN()))}    

*part of   equation7 



.FUNC ALPHAF() {((Q*A)/IES())*(F1()-(F3()*((NOB*WB)/(2*TAUB))))}

.FUNC BF() {ALPHAF()/(1-ALPHAF())}

.FUNC F4()  {(NOB*WB/(2*TAUB))}



 

.FUNC F5()  {(SCN()+SCN()*SEN())/(SEN()+SCN()+SCN()*SEN())}      

* second part of equation 7

.FUNC ICS() {Q*A*(F1()+F4()*F5()+(DC*POC/(WC*(1+1/SCP()))))}    * equation  1

.PARAM OF=.001


*optical feedback parameter*

.PARAM EF=.01



*electrical feedback parameter*

.PARAM EE=10



*early effect parameter*


* The circuit


IIN 0 4 DC .2mA

G3 4 3 VALUE={OF*(I(VCC))}

G4 0 4 VALUE={EF*((I(VCC))^(1.5))}

R 3 0 1k

R1 1 2 100

VI 4 5 DC 0

G1 3 0 VALUE={EE*V(3,0)*I(VI)}

VCC 1 0 DC 5

Q1 3 5 0 QA

.MODEL QA NPN (ISE={IES()} ISC={ICS()} BF={BF()} VJE=1.51 VJC=1.36)

D1 2 3 DA

.MODEL DA D(IS=1E-10 VJ=1.3 RS=50 N=1.5 )

.DC VCC 0 10V .1V IIN 0 .2mA 20UA

.PROBE

.END
Fig.3. PSPICE input file for LAOS.

In Fig. 4, the I-V characteristics of the LAOS under low-bias condition are shown.  These characteristics are identical to an HPT.  However the switching from low current high resistance to high current low resistance via a negative differential resistance occurred at large bias as shown in Fig. 5.  This behavior is in agreement with theoretical and experimental results in [8] and [13], which proves the accuracy of the model presented here.

To further show the usefulness of the model, the effect of some parameters on the break over voltage VBO, that is the voltage at which the device switch from low current state to high current state, is presented next.  Figures 6 and 7 show that increasing the base doping as well as the base width has the effect of increasing the break over voltage which is probably due to the decrease in the gain. Therefore, these parameters should be carefully chosen in order to improve the switching behavior of the device.

In contrast, increasing the base current as well as the emitter doping causes the device to switch at lower voltage as shown in Figs. 8 and 9. This can be attributed to the increase in the injection efficiency as the emitter doping increases which improves 

the gain, while increasing the base current drives the HPT toward saturation and hence reduce the voltage at which the device switches.

Similarly, the effect of each of optical feedback, electrical feedback, and the Early effect shows the same behavior. That is, increasing such parameters causes the breakover voltage to decrease as shown in Figs. 10,11 and 12, respectively in agreement with published data [8,13]

Finally the effect of the junction velocity (SEN), which is a function of the conduction band discontinuity ((EC) between the emitter and the base (see Eq. 6) is represented in Fig. 13.  In this figure, it can be seen that increasing SEN has the effect of making the device to switch at a lower voltage.

Conclusion

A PSPICE model for the Light Amplifying Optical Switch (LAOS) is presented.  The effect of some parameters on the breakover voltage are shown.  The model is simple and can be incorporated in the educational environment in order to simulate more complicated optical systems using LAOS.

تحليل الحالة الساكنة للمفتاح المكبر للضوء باستخدام برنامج سبايس
حمد صالح الحقيل
قسم الهندسة الكهربائية ، كلية الهندسة ، جامعة الملك سعود، ص. ب. 800 
 الرياض 11421 ، المملكة العربية السعودية
 ( قدم للنشر في 25/07/2001م؛وقبل للنشر في 16/03/2002م )
ملخص البحث. تم من قبل بعض الباحثين إيجاد مفتاح بصري مكبرا للضوء. يتكون هذا المفتاح من ترانزيستور ضوئي ذي وصلة متباينة التجانس تتكامل تعاقبيا مع ثنائي باعث للضوء. كما وجد أن هذا المفتاح قادر على التكبير كما أنه ينتقل من منطقة تيار منخفض إلى منطقة تيار مرتفع – مرحلة الغلق والفتح- وذلك عندما تصل الفولتية المسلطة عليه إلى نقطة معينه سميت بـنقطة التحول الانكساري. وأرجع السبب في هذا الانتقال إلى التغذية الخلفية الضوئية والكهربائية بالإضافة إلى عدم خطية الكسب. وتم إيجاد دائرة مكافئة لهذا المفتاح مع الأخذ في الاعتبار العوامل المؤثرة في انغلاقه وانفتاحه وذلك من قبل بعض الباحثين. وفي هذه الدراسة تم إجراء تعديل على الدائرة السابق نشرها ومن ثم تحليل هذا المفتاح باستخدام برنامج سبايس الشائع في البيئة التعليمية ووجد تطابق نوعي بين النتائج المقدمة هنا مع تلك المنشورة من قبل باحثين آخرين. كما تمت دراسة العوامل المؤثرة على نقطة التحول الانكساري وكانت النتائج مطابقة لما نشر. وعلى هذا فإن نموذج سبايس المقدم هنا سوف يساعد الطلاب على دراسة دوائر ونظم الإلكترونيات البصرية والتي تستخدم هذا المفتاح نظرا لشيوع هذا البرنامج في البيئات التعليمية مما سيكون له الأثر في فهم هذه الدوائر والنظم والعوامل المؤثرة عليها.

Fig. 13.  The breakover voltage as a function of emitter base junction velocity.

Fig.8. Effect of emitter doping on the breakover voltage.

Fig. 9. Effect of input current on the breakover voltage. Note that the input current is directly proportional to the input light.

Fig. 12. The breakover voltage as a function of early effect parameter.

Fig. 10. The breakover voltage as a function of optical feedback parameter.

Fig. 11. The breakover voltage as a function of electrical feedback parameter.

Fig.1. (a) Physical structure of the LAOS; (b) Equivalent circuit of the LAOS.	

Fig.2. The equivalent circuit of the LAOS used in simulation. Note: I is the total Laos current,  is the optical feedback coefficient, χ is the electrical feedback coefficient and  is the early effect coefficient.

Fig.4. LAOS I-V characteristics at low bias.

Fig.5. LAOS I-V characteristics at large bias. Note the switching from low–current to high-current state.

Fig.6. Effect of base doping on the breakover voltage.

Fig.7. Effect of base width on the breakover voltage.

